Rechnerstrukturen, Teil 2

2014/06/04

Kontext

Die Wissenschaft Informatik befasst sich mit der Darstellung, Speicherung, Übertragung und Verarbeitung von Information
[Gesellschaft für Informatik]

Z.B. Zahlendarstellung in RS, Teil 1

hier und jetzt

Motivation

- Jede Ausführung von Programmen bedarf einer zur Ausführung fähigen Hardware.
- Wir nennen diese auch Ausführungsplattformen (***execution platforms)***.
- *Platform-based design* ist ein Ansatz für viele Anwendungen (Handys, Autos, …)
- Plattformen sind nicht immer ideal (z.B. führen Anwendungen nicht in 0 Zeit mit 0 Energie aus)
- Grundlegendes Verständnis für nicht-ideales Verhalten ist wichtig
- Deshalb Beschäftigung in dieser Vorlesung mit Ausführungsplattformen.



Problematische Situationen …



… und Techniken zu deren Vermeidung



Thema des zweiten Teils der Vorlesung

Ziel: Verständnis der Arbeitsweise von Rechnern, einschl.

- der Programmierung von Rechnern
- des Prozessors
- der Speicherorganisation
- des Anschlusses von Peripherie
- Anwendungen bei Eingebetteten Systemen

Stil des zweiten Teils der Vorlesung

Stil:

- Betonung des Skripts
- Integration mit praktischen Übungen
  Einsatz eines Simulators

- Buch von Hennessy/Patterson: *Computer Organization: The hardware/software interface*, Morgan Kaufman, 2. Aufl. (siehe Lehrbuchsammlung) oder 3. Aufl. als grobe Leitlinie
- Weitere Bücher: siehe Literaturverzeichnis im Skript

- Soviel Interaktion, wie möglich



Rechnerarchitektur
- Einleitung -

Abstraktionsebenen:



2.2 Die Befehlsschnittstelle
2.2.1 Der MIPS-Befehlssatz

Beispiel: MIPS (~ *machine with no interlocked pipe stages*)

≠ MIPS (*million operations per second*)
Entwurf Anfang der 80er Jahre

Warum MIPS ?

- Weitgehend sauberer und klarer Befehlssatz
- Kein historischer Ballast
- Basis der richtungsweisenden Bücher von Hennessy/Patterson
- Simulator verfügbar
- MIPS außerhalb von PCs (bei Druckern, Routern, Handys) weit verbreitet

MIPS R2000
Microprocessor without interlocked pipeline stages



Begriffe

Assemblerprogramm-Ebene
Maschinenprogramm-Ebene / Befehlsschnittstelle

- **MIPS-Befehle** sind elementare Anweisungen an die MIPS-Maschine
- Ein **MIPS-Maschinenprogramm** ist eine konkrete Folge von MIPS-Befehlen
- Die **MIPS-Maschinensprache** ist die Menge möglicher MIPS-Maschinenprogramme
- Entsprechendes gilt für die Assemblerprogramm-Ebene
- Vielfach keine Unterscheidung zwischen beiden Ebenen

MIPS-Assembler-Befehle
Arithmetische und Transportbefehle



Erstes Beispiel: Addition:

Allgemeines Format:

`add a,b,c` mit `a,b,c` ∈ `$0..$31`

`$0..$31` stehen für 32 Bit lange Register,

`a` ist das Zielregister,

Beispiel:

```
add $3,$2,$1
```

Funktion des Additionsbefehls

Beispiel: **`add $3,$2,$1`**

Register („**`Reg`**“)



`$0` ist ständig = 0

`add $3,$2,$1`

Simulation des MIPS-Befehlssatzes mit MARS

Simulator MARS: www.cs.missouristate.edu/MARS/ (erfordert Java),

Warum Simulation statt Ausführung des Intel-Befehlssatzes?

- Möglichkeit der Benutzung des MIPS-Befehlssatzes
- Keine Gefährdung des laufenden Systems
- Bessere Interaktionsmöglichkeiten
- MARS läuft auf verschiedenen Plattformen: Windows XP, Vista, Windows 7, MAC OS X, Linux

Installation jetzt dringend empfohlen, für Studierende ohne PC-Zugang Nutzung an Fakultätsrechnern möglich.

Schreiben Sie eigene kleine Programme!

Üblicherweise beziehen sich mindestens 2 von 4 Klausuraufgaben auf die MIPS-Programmierung.

Älterer Simulator: SPIM

- MARS versucht, SPIM-Obermenge zu implementieren
- Bekannte Ausnahmen:
  - .set-Anweisung
  - *Delayed branches*
  - Kein Standard-*trap handler* in MARS
  - Keine Extra-Befehlsliste verfügbar
  - Programm-Ende mittels exit-syscall statt mit **`jr $31`**
  - Andere Realisierung großer Konstanten
- (SPIM-) Befehlsliste befindet sich im Anhang des Skripts.

Semantik: per Register-Transfer-Notation
(genauer: Register-Transfer-Verhaltens-Notation)

Argumente oder Ziele: Register oder Speicher, z.B.

```
Reg[3]; PC;  Reg[31]
```

Zuweisungen: mittels `:=` , z.B.

```
PC := Reg[31]
```

Konstante Bitvektoren: Einschluss in ", z.B:

```
"01010100011"
```

Selektion von einzelnen Bits: Punkt + runde Klammern:

```
PC.(15:0)
```

Konkatenation (Aneinanderreihung) mit `&`, z.B.

```
(Hi & Lo);
PC := PC.(31:28) & I.(25:0) & "00"
```

Semantik des Additionsbefehls

Bedeutung in Register-Transfer-Notation

Beispiel:

```
add $3,$2,$1        # Reg[3]:= Reg[2]+Reg[1]
```

leitet in der Assemblernotation einen Kommentar ein.

Register speichern (Teils des) aktuellen Zustands;
`add`-Befehl veranlasst Zustandstransformation.

Addition für 2k-Zahlen oder für vorzeichenlose Zahlen? (1)

Muss der **`add`**-Befehl „linkestes“ Bit als Vorzeichen betrachten?

Nein, bei 2k-Zahlen und bei vorzeichenlosen Zahlen („Betragszahlen“) kann jede Stelle des Ergebnisses gemäß der Gleichungen für Volladdierer bestimmt werden, unabhängig davon, ob die Bitfolgen als 2k-Zahlen oder als Betragszahlen zu interpretieren sind (siehe RS, Teil 1).

- ☞ es reicht ein Additionsbefehl zur Erzeugung der Ergebnis-Bitvektoren für beide Datentypen aus.
  Allerdings Unterschiede hinsichtlich der Behandlung von Bereichsüberschreitungen.

Addition für 2k-Zahlen oder für vorzeichenlose Zahlen? (2)

MIPS-Architektur bietet **`addu`**-Befehl für *unsigned integers:*

- **`add`**-Befehl signalisiert Bereichsüberschreitungen (für vorzeichenbehaftete Zahlen),
- **`addu`**-Befehl ignoriert diese (kein Signalisieren von Bereichsüberschreitungen vorzeichenloser Zahlen).

Das Speichermodell der MIPS-Architektur



Eigenschaften des Von-Neumann-Rechners (1)

Wesentliches Merkmale der heute üblichen Rechner, die auf dem Prinzip des Von-Neumann-Rechners basieren:

1. Einteilung des Speichers in Zellen gleicher Größe, die über **Adressen** angesprochen werden können.



Der `load word`-Befehl

Allgemeine Form:
`lw ziel, offset(reg)` mit **`ziel, reg`** $\in$ **`$0..$31,`**
`offset:` Konstante $\in -2^{15} .. 2^{15}-1$, deren Wert beim Laden des Programms bekannt sein muss bzw. deren Bezeichner.
Beispiel:

```
lw $8, 12($19)# Reg[8]:=Speicher[12+Reg[19]]
```



Der `load word`-Befehl

Allgemeine Form:
`lw ziel, offset(reg)` mit **`ziel, reg`** $\in$ **`$0..$31,`**
`offset:` Konstante $\in -2^{15} .. 2^{15}-1$, deren Wert beim Laden des Programms bekannt sein muss bzw. deren Bezeichner.
Beispiel:

```
lw $8, 12($19)# Reg[8]:=Speicher[12+Reg[19]]
```



Der `store word`-Befehl

Allgemeine Form:
`sw quel, offset(reg)` mit **`quel, reg`** $\in$ **`$0..$31`**,
`offset:` Konstante $\in -2^{15} .. 2^{15}-1$, deren Wert beim Laden des Programms bekannt sein muss bzw. deren Bezeichner.
Beispiel:

```
sw $8, 8($19) # Speicher[8+Reg[19]] := Reg[8]
```



Der `store word`-Befehl

Allgemeine Form:
`sw quel, offset(reg)` mit **`quel, reg ∈ $0..$31,`**
`offset:` Konstante ∈ -2^{15} .. $2^{15}-1$, deren Wert beim Laden des Programms bekannt sein muss bzw. deren Bezeichner.
Beispiel:
`sw $8, 8($19) # Speicher[8+Reg[19]] := Reg[8]`



Darstellung von Befehlen im Rechner

Zergliederung eines Befehlswortes in **Befehlsfelder**;
Jede benutzte Zergliederung heißt **Befehlsformat;**
Beispiele:



	Größe [Bit]	6	5	5	5	5	6
Arithmetische Befehle	Format R	op	rs	rt	rd	shamt	funct
`lw, sw`	Format I	op	rs	rt	offset		

Speicherung von Befehlen im Rechner

Assemblerprogramm

```
lw   $2,100($0)
lw   $3,104($0)
add $3,$2,$3
sw   $3,108($0)
```

Übersetzung des Assemblerprogramms in ein ladbares Maschinenprogramm ist Aufgabe des **Assemblers**

Hauptspeicher (**`Speicher`**)



Eigenschaften des Von-Neumann-Rechners (2)

Wesentliche Merkmale der heute üblichen Rechner, die auf dem Prinzip des Von-Neumann-Rechners basieren:

2. Verwendung von speicherprogrammierbaren Programmen.

3. Speicherung von Programmen und Daten in demselben Speicher.



Zusammenfassung

- Schichtenmodell
  - Programme in höherer Programmiersprache
  - Assemblerprogramme
  - Maschinenprogramme
  - RT-Verhalten/-Strukturen
  - Gatter
- Unterscheidung zw. Befehlen, Programmen, Sprachen
- Exemplarische Betrachtung der MIPS-Assembler- & Maschinensprache ☞ MARS
  - add-, lw-, sw-Befehle; RT-Semantik; Unterscheidung add/addu
  - Speichermodell
  - Darstellung von Befehlen
- Prinzipien der von Neumann-Maschine

Abarbeitung: immer der Reihe nach

Reg

$0	
$1	
$2	15
$3	10
..	0

Hauptspeicher (**Speicher**)

0	8c 02 00 64	lw $2,100($0)
4	8c 03 00 68	lw $3,104($0)
8	00 43 18 20	add $3,$2,$3
12	9c 03 00 6c	sw $3,108($0)
...		
100	15	
104	10	
108	25	
...		
$\leq 2^{32}-4$		

Der Zeiger auf den gerade ausgeführten Befehl wird im Programmzähler **PC** (*program counter*) gespeichert

Eigenschaften des Von-Neumann-Rechners (3)

Wesentliche Merkmale der heute üblichen Rechner, die auf dem Prinzip des Von-Neumann-Rechners basieren:

4. Die sequentielle Abarbeitung von Befehlen.



5. Es gehört zur Semantik eines jeden Befehls (Ausnahme: Sprungbefehle, s.u.), dass zusätzlich zu den erwähnten Änderungen der Speicherinhalte noch der Programmzähler PC erhöht wird, im Fall der MIPS-Maschine jeweils um 4.



2.2.1 Der MIPS-Befehlssatz
2.2.1.1 Arithmetische und Transportbefehle

Darstellung von Befehlen im Rechner (Wdh.)

	Größe [Bit]	6	5	5	5	5	6
Arithmetische Befehle	Format R	op	rs	rt	rd	shamt	funct
`lw, sw`	Format I	op	rs	rt	offset		

Problem: Passt nicht zur Länge der Register

Nutzung des 16-Bit-Offsets in der 32-Bit Arithmetik

Replizierung des Vorzeichenbits liefert 32-Bit 2k-Zahl:



`sign_ext(a,m)` : Funktion, die Bitvektor **`a`** auf **`m`** Bits erweitert

Beweis der Korrektheit von `sign_ext`

Bitvektor a=(a_{n-1}, …,a_0) repräsentiert bei 2k-Kodierung Zahl

$$\text{int}(a) = -a_{n-1}2^{n-1} + \sum_{i=0}^{n-2} a_i 2^i$$

- Sei nun a_{n-1}='0'. Dann ist

$$\text{int}(\text{sign\_ext}(a,m)) = \text{int}("\overbrace{00..0}^{m-n+1}"\&(a_{n-2},..a_0)) = \sum_{i=0}^{n-2} a_i 2^i = \text{int}(a)$$

- Sei nun a_{n-1}='1'. Dann ist

$$\text{int}(\text{sign\_ext}(a,m)) = \text{int}("\overbrace{11..1}^{m-n+1}"\&(a_{n-2},..a_0)) = -2^{m-1} + \sum_{i=n-1}^{m-2} 2^i + \sum_{i=0}^{n-2} a_i 2^i$$

Wegen $\sum_{i=n-1}^{m-2} 2^i + 2^{n-1} = 2^{m-1}$ folgt $-2^{m-1} + \sum_{i=n-1}^{m-2} 2^i = -2^{n-1}$, also auch

$$\text{int}(\text{sign\_ext}(a,m)) = -2^{n-1} + \sum_{i=0}^{n-2} a_i 2^i = \text{int}(a)$$ q.e.d.

Auswirkung der Nutzung von `sign_ext` bei der Adressierung

Präzisere Beschreibung der Bedeutung des sw-Befehls:

`sw rt, offset(rs) #`

`Speicher[Reg[rs]+sign_ext(offset,32)]:=Reg[rt]`

mit Vorzeichenweiterung



mit *zero-extend* (“00..00” & ...)



Einsatz der Vorzeichenerweiterung bei Direktoperanden (*immediate addressing*)

- Beschreibung der Bedeutung des *add immediate*-Befehls*

  `addi rt, rs, const #` benutzt Format I

  `# Reg[rt] := Reg[rs] + sign_ext(const,32)`

- *add immediate unsigned*-Befehl

  `addiu rt, rs, const #` benutzt Format I

  Wie **`addi`** ohne Erzeugung von Überläufen **(!)**

* Der MIPS-Assembler erzeugt vielfach automatisch *immediate*-Befehle, wenn statt eines Registers direkt eine Konstante angegeben ist, z.B. bei **`add $3,$2,3`**

Subtraktionsbefehle



```
sub $4,$3,$2 # Reg[4]:= Reg[3]-Reg[2]
```

Subtraktion, Ausnahmen möglich

```
subu $4,$3,$2 # Reg[4]:= Reg[3]-Reg[2]
```

Subtraktion, keine Ausnahmen signalisiert

Format: R

Multiplikationsbefehle

Die Multiplikation liefert doppelt lange Ergebnisse.

Beispiel: $-2^{31} \times -2^{31} = 2^{62}$;

2^{62} benötigt zu Darstellung einen 64-Bit-Vektor.

Wo soll man ein solches Ergebnis abspeichern?

MIPS-Lösung: 2 spezielle Register **`Hi`** und **`Lo:`**

```
mult $2,$3 # Hi & Lo := Reg[2] * Reg[3]
```

Höherwertiger Teil des Ergebnisses

Niederwertiger Teil des Ergebnisses

Konkatenation (Aneinanderreihung)

Transport in allgemeine Register:

```
mfhi, $3 # Reg[3]:=Hi
mflo, $3 # Reg[3]:=Lo
```

Varianten des Multiplikationsbefehls

- **`mult $2,$3 # Hi&Lo:=Reg[2]*Reg[3];`**
  für ganze Zahlen in 2k-Darstellung
- **`multu $2,$3 # Hi&Lo:=Reg[2]*`$_u$ **`Reg[3];`**
  für natürliche Zahlen (*unsigned int)*
- **`mul $4,$3,$2`** # Besteht aus **`mult`** und **`mflo`**
  `# Hi&Lo:=Reg[3]*Reg[2]; Reg[4]:=Lo;`
  für 2k-Zahlen, niederwertiger Teil im allgem. Reg.
- **`mulo $4,$3,$2`** #
  `# Hi&Lo:=Reg[3]* Reg[2]; Reg[4]:=Lo;`
  für 2k-Zahlen, niederwertiger Teil im allg. Reg, Überlauf-Test.
- **`mulou $4,$3,$2`** #
  `# Hi&Lo:=Reg[3]*`$_u$ **`Reg[2]; Reg[4]:=Lo;`
  für *unsigned integers*, im allg. Reg., Überlauf-Test

Merkregel:
„weniger ist mehr“
= kürzere Bezeichung
kopiert auch in allgem. Register

Divisionsbefehle



Problem: man möchte gern sowohl den Quotienten wie auch den Rest der Division speichern;
Passt nicht zum Konzept eines Ergebnisregisters

MIPS-Lösung: Verwendung von **`Hi`** und **`Lo`**

- **`div $2,$3 #`** für ganze Zahlen in 2k-Darstellung
  `# Lo:=Reg[2]/Reg[3]; Hi:=Reg[2] mod Reg[3]`

- **`divu $2,$3 #`** für natürliche Zahlen (*unsigned integers*)
  `# Lo:=Reg[2]/`$_{\mathbf{u}}$ **`Reg[3]; Hi:=Reg[2] mod Reg[3]`**

Logische Befehle

Beispiel	Bedeutung	Kommentar
`and $4,$3,$2`	`Reg[4]:=Reg[3] ∧ Reg[2]`	und
`or $4,$3,$2`	`Reg[4]:=Reg[3] ∨ Reg[2]`	oder
`andi $4,$3,100`	`Reg[4]:=Reg[3] ∧ 100` (zero_ext( )	und mit Konstanten
`sll $4,$3,10`	`Reg[4]:=Reg[3] << 10`	Schiebe nach links logisch
`srl $4,$3,10`	`Reg[4]:=Reg[3] >> 10`	Schiebe nach rechts logisch

Laden von Konstanten

Wie kann man 32-Bit-Konstanten in Register laden?

- Direktoperanden für das untere Halbwort:

```
ori    r,s,const   # Reg[r]:=Reg[s] ∨ (0000_16 & const)
addiu  r,s,const   # Reg[r]:=Reg[s] + sign_ext(const,32)
```

- Für den Sonderfall s=$0:

```
ori r,$0,const     #  Reg[r]:=0 ∨ zero_ext(const,32)
addiu r,$0,const   #  Reg[r]:=sign_ext(const,32)
```





Laden von Konstanten (2)

- Für Konstanten mit: unteres Halbwort = 0
  `lui r, const` **#**Reg[r]:=const &0000_{16}
  (*load upper immediate*)

const	0

Laden von Konstanten (3)

- Für andere Konstanten:
  `lui $1, const‘`
  `ori  r,$1,const“`



Sequenz wird vom Assembler für **`li`** (*load immediate*) erzeugt.

Unterschiedliche Abbildung auf Maschinenbefehle bei SPIM und MARS!

Register $1 ist immer für den Assembler freizuhalten.

Der load address – Befehl `la`

In vielen Fällen muss die Adresse einer Speicherzelle in einem Register bereit gestellt werden.

Dies ist mit den bislang vorgestellten Befehlen zwar möglich, der Lesbarkeit wegen wird ein eigener Befehl eingeführt.

Der Befehl **`la`** entspricht dem **`lw`**-Befehl,
wobei der Speicherzugriff unterbleibt. Beispiel:

```
la $2,0x20($3) # Reg[2] := 0x20 + Reg[3]
```

Einteilung des Speicherbereichs (*memory map*) im Simulator folgt üblicher C/Unix-Konvention



Befehle

Daten bekannter Größe+weitere Daten

Dynamisch erzeugte Daten, Auf- und Abbau bei Aufruf von Prozeduren/Unterprogrammen

Adresse Ein-/ Ausgabegeräte

Betriebssystem

Problem der Nutzung nicht zusammenhängender Adressbereiche



Man müsste den Segmenten verschiedenen physikalischen Speicher zuordnen, wenn man mit diesen Adressen wirklich den physikalischen Speicher adressieren würde.

Auswege:

- Umrechnung der Adressen (siehe Abschnitt über Speicher) oder
- Benutzung weitgehend zusammenhängender Speicherbereiche

Benutzung weitgehend zusammenhängender Speicherbereiche

Betriebssystem

Text segment

Data segment

Stack segment

....

Erfordert eine relativ gute Kenntnis der Größe der Segmente

Kann beim MARS konfiguriert werden (☞ Einstellungen).

Benutzung der Speicherbereiche in einem Beispielprogramm

```
Beispielprogramm
        .glob main                      #Globales Symbol
main: lw $2, 0x10010000($0)             #Anfang Datenbereich
        lw $3, 0x10010004($0)
        add $3,$2,$3
        sw $3, 0x10010008($0)
```

MARS erzeugt im Fall „zu großer" Offsets in einem Befehl

```
        lw $z, d($b)
```

Code zur Berechung der Adresse nach dem folgenden Schema in $1

```
        lui $1,<obere 16 Bit von d>
        addu $1,$1,$b; ohne overflow check
        lw $z,<untere 16 Bit von d>($1)
```

Der 2. Befehl entfällt wenn ($b) nicht vorhanden ist, aber nicht bei b=0 (!)

Zwei Versionen des Additionsprograms

```
        .globl main                    #Globales Symbol
main: lw $2, 0x10010000($0)            #Anfang Datenbereich
      lw $3, 0x10010004($0)
      add $3,$2,$3
      sw $3, 0x10010008($0)
       … syscall
```

```
        .globl main
main:  li $28,0x10010000               # Anfang Datenbereich
       lw $2, 0($28)                   # in einem Register
       lw $3, 4($28)                   # gespeichert
       add $3,$2,$3
       sw $3, 8($28)
       … syscall
```

Transformation der Programmdarstellungen



Funktion des Assemblers (1)

- Übersetzt symbolische Befehlsbezeichnungen in Bitvektoren.
- Übersetzt symbolische Registerbezeichnungen in Bitvektoren.
- Übersetzt Pseudo-Befehle in echte Maschinenbefehle.
- Verwaltet symbolische Marken.
- Nimmt die Unterteilung in verschiedene Speicherbereiche vor.

Funktion des Assemblers (2)

- Verarbeitet einige Anweisungen an den Assembler selbst:

.ascii ***text*** Text wird im Datensegment abgelegt

.asciiz ***text*** Text wird im Datensegment abgelegt, 0 am Ende

.data die nächsten Worte sollen in das Daten-Segment

.extern Bezug auf externes globales Symbol

.globl *id* Bezeichner *id* soll global sichtbar sein

.kdata die nächsten Worte kommen in das Kernel-Daten-Segment

.ktext die nächsten Worte kommen in das Kernel-Text-Segment

.set Setzen von Optionen (SPIM)

.space ***n*** ***n*** Bytes im Daten-Segment reservieren

.text die nächsten Worte kommen in das Text-Segment

.word *wert, ... wert* Im aktuellen Bereich zu speichern

Assembler übersetzt symbolische Registernummern

$zero = $0 usw., siehe Anhang

zero 0 Constant 0	s0 16 Saved temporary, preserved across call
at 1 Reserved for assembler	s1 17 Saved temporary, preserved across call
v0 2 Expression evaluation and	s2 18 Saved temporary, preserved across call
v1 3 results of a function	s3 19 Saved temporary, preserved across call
a0 4 Argument 1	s4 20 Saved temporary, preserved across call
a1 5 Argument 2	s5 21 Saved temporary, preserved across call
a2 6 Argument 3	s6 22 Saved temporary, preserved across call
a3 7 Argument 4	s7 23 Saved temporary, preserved across call
t0 8 Temporary, ¬ preserved across call	t8 24 Temporary, not preserved across call
t1 9 Temporary, ¬ preserved across call	t9 25 Temporary, not preserved across call
t2 10 Temporary, ¬ preserved across call	k0 26 Reserved for OS kernel
t3 11 Temporary, ¬ preserved across call	k1 27 Reserved for OS kernel
t4 12 Temporary, ¬ preserved across call	gp 28 Pointer to global area
t5 13 Temporary, ¬ preserved across call	sp 29 Stack pointer
t6 14 Temporary, ¬ preserved across call	fp 30 Frame pointer
t7 15 Temporary, ¬ preserved across call	ra 31 Return address, used by function call

Zusammenfassung

- Sequentielle Befehlsbearbeitung
- Vorzeichenerweiterung
- Laden von Konstanten
- Weitere arithmetische/logische Befehle
- Übliche Speichereinteilung
- Funktion des Assemblers

2.2.1.5 Sprungbefehle

Problem mit dem bislang erklärten Befehlssatz:
keine Abfragen möglich.

Lösung: (bedingte) Sprungbefehle (*conditional branches*).

Elementare MIPS-Befehle:

`beq rega,regb,Sprungziel` (*branch if equal*)
mit **`rega, regb`** $\in$ **`$0..$31`**,
und **`Sprungziel`**: 16 Bit *integer* (oder Bezeichner dafür).

`bne rega,regb,Sprungziel` (*branch if not equal*)
mit **`rega, regb`** $\in$ **`$0..$31`**,
und **`Sprungziel`**: 16 Bit *integer* (oder Bezeichner dafür).

Anwendung von `beq` und `bne`

Übersetzung von

```
if (i==j) goto L1;
f=g+h;
L1: f=f-i;
```

Assembler rechnet L1 in die im Maschinencode zu speichernde Konstante um.

in

```
beq $19, $20, L1          # nach L1, falls i=j
    add $16, $17, $18     # f=g+h
L1: sub $16, $16, $19     # immer ausgeführt
```

Symbolische Bezeichnung für eine Befehlsadresse.

Unbedingte Sprünge

Problem: Übersetzung von

```
if (i==j) f=g+h; else f = g -h;
```

Lösung

```
bne $19, $20, Else          # nach Else falls i≠j
    add $16, $17, $18       # f=g+h
    j   Exit                # Unbedingter Sprung
                            # (unconditional jump)
Else: sub $16, $17, $18     # f=g-h
Exit: ...
```

Realisierung von Array-Zugriffen



- Bei C beginnen Arrays mit dem Index 0
- Die Adresse des Arrays ist gleich der Adresse der Komponente 0
- Wenn jede Array-Komponente ein Wort belegt, dann belegt Komponente i das Wort i des Arrays.
- Wenn a_save die Anfangsadresse eines Arrays ist, dann ist (a_save+ i x c) die Adresse der Komponente i.
  c ist die Anzahl der adressierbaren Speicherzellen, die pro Element des Arrays belegt werden (=4 bei 32-Bit Integern auf der MIPS-Maschine)

Realisierung von Schleifen mit unbedingten Sprüngen



Problem: Übersetzung von

```
while (save[i]==k) i = i+j;
```



Lösung:

```
        li   $10,4            # Reg[10]:=4
Loop:   mul  $9,$19,$10       # Reg[9]:=i * 4
        lw   $8,a_save($9)    # Reg[8]:=save[i]
        bne  $8,$21, Exit     # nach Exit, falls ≠
        add  $19,$19,$20      # i=i+j
        j    Loop
Exit:   ...
```

Format und Bedeutung unbedingter Sprünge

	Größe [Bit]	6	5	5	5	5	6
Arithmetische Befehle	Format R	op	rs	rt	rd	shamt	funct
`lw, sw`	Format I	op	rs	rt	offset		
`j, jal`	Format J	op	adresse				



Tests auf <, ≤ ,>, ≥



MIPS-Lösung: **`slt`**-Befehl (*set if less than*)

```
slt ra,rb,rc  #
# Reg[ra]:=if Reg[rb]<Reg[rc] then 1 else 0
```

Tests werden vom Assembler aus **`slt, bne`** und **`beq`**-Befehlen zusammengesetzt:

Beispiel:
aus **`blt $2,$3,L`** wird

```
slt $1,$2,$3
bne $1,$0,L
```

`$1` **ist per Konvention dem Assembler vorbehalten**

Weitere Befehle

Weitere Befehle:

- slti (Vergleich mit Direktoperanden)
- sltu (Vergleich für Betragszahlen)
- sltui (Vergleich für Betragszahlen als Direktoperand)
- ble (Verzweige für *less or equal*)
- blt (Verzweige für *less than*)
- bgt (Verzweige für *greater than*)
- bge (Verzweige für *greater or equal*)

} Pseudo-befehle

Übersetzung einer `for`-Schleife

Bedingte Sprünge werden auch zur Übersetzung von **`for`**-Schleifen benötigt. Beispiel: Das C-Programm

```
j = 0;
for (i=0; i<n; i++)  j = j+i;
```

kann in das folgende Programm übersetzt werden:

```
main:  li   $2,0              # j:=0
       li   $3,0              # i:=0
loop:  bge  $3,$4, ende       # i>=n?
       add  $2,$2,$3          # j=j+i
       addi $3,$3,1           # i=i+1
       j    loop
ende:  ...
```

Realisierung von berechneten Sprüngen: der `jr`-Befehl

Format: **`jr reg`** (jump register), mit **`reg ∈ $0..$31`**

Semantik: **`PC:=Reg[reg]`**

Beispiel: **`jr $3`**

Registerspeicher (**`Reg`**)

`$0`				
`$..`				
`$3`	00	40	00	00_{16}
`$..`				
`$30`				
`$31`				

PC



SWITCH-Anweisungen

Annahme: k=2

```
switch(k) { /* k muss ∈ 0..3 sein! */
 case 0: f=i+j; break; /* k =0 */
 case 1: f=g+h; break; /* k =1 */
 case 2: f=g-h; break; /* k =2 */
 case 3: f=i-j; break; /* k =3 */
}
```




```
.data
Jumptable: .word L0,L1,L2,L3, ..
```

Realisierung von SWITCH-Anweisungen mittels des jr-Befehls

```
switch(k) {
 case 0: f=i+j; break; /* k =0 */
 case 1: f=g+h; break; /* k =1 */
 case 2: f=g-h; break; /* k =2 */
 case 3: f=i-j; break; /* k =3 */ }
```

```
       li   $10,4                 # Reg[10]:=4
swit:  mul  $9,$10,$21            # Reg[9]:=k*4
       lw   $8,Jumptable($9)      # 1 der 4 Adressen
       jr   $8                    # PC:=Reg[8]
L0:    add  $16,$19,$20           # k ist 0;
       j    Exit                  # entspricht break
L1:    add  $16,$17,$18           # k ist 1;
       j    Exit                  # entspricht break
L2:    sub  $16,$17,$18           # k ist 2;
       j    Exit                  # entspricht break
L3:    sub  $16,$19,$20           # k ist 3;
Exit:  ...                        #
```

Zusammenfassung

- Bedingte Sprungbefehle
- (Unbedingte) Sprungbefehle
- Realisierung von for-Schleifen
- Realisierung von switch-Anweisungen

2.2.1.6 Prozeduraufrufe

```
void function C
{
  ...
}
void function B
{
C
}
void function A
{
B
}
main: A
```

Man muss

- sich merken können, welches der auf den aktuellen Befehl im Speicher folgende ist (d.h., man muss sich PC+4 merken) und
- an eine (Befehls-) Adresse springen.

Der *jump-and-link*-Befehl

Format: **`jal adresse`**

wobei **`adresse`** im aktuellen 256 MB-Block liegt

Bedeutung: **`Reg[31]:=PC+4; PC:=adresse`**

Beispiel: **`jal 0x410000`**

Registerspeicher (**`Reg`**)

`$0`				
`$1`				
`$..`				
`$..`				
`$30`				
`$31`	00	40	00	04_{16}

Nach der Ausführung von **jal** an Adresse 400000_{16}

PC

00	41	00	00_{16}

Realisierung nicht verschachtelter Prozeduraufrufe mit `jal` und `jr`



Realisierung nicht verschachtelter Prozeduraufrufe mit `jal` und `jr`



Das Stapel-Prinzip

```
void function C
{
  ...
}
void function B
{
C
}
void function A
{
B
}
main: A
```

Speicherbereich für (1. Aufruf) von C
Speicherbereich für (1. Aufruf) von B
Speicherbereich für (1. Aufruf) von A
Evtl. Speicherbereich für Hauptprogramm

Realisierung eines Stapels im Speicher



MIPS-Konvention:
Benutzung von Register 29 als *stack-pointer*

Konkrete Realisierung für drei Prozeduren

Statisch:

```
C()
  { ...; }    /* jr*/
B()
  {          /*$31->stack*/
   C();
  }          /*stack->$31,jr */
A()
  {          /*$31->stack*/
   B();
  }         /*stack->$31,jr */
main()
  {A();}
```

Dynamisch:

Aktiv	Befehl
main	jal A
A	$31 -> stack
A	jal B
B	$31 -> stack
B	jal C
C	jr $31
B	Rückschreiben von $31
B	jr $31
A	Rückschreiben von $31
A	jr $31
main	

Sichern von Registerinhalten



Einige Register sollten von allen Prozeduren genutzt werden können, unabhängig davon, welche Prozeduren sie rufen und von welchen Prozeduren sie gerufen werden.

Die Registerinhalte müssen beim Prozeduraufruf gerettet und nach dem Aufruf zurückgeschrieben werden.

2 Methoden:

1. Aufrufende Prozedur rettet vor Unterprogrammaufruf Registerinhalte (*caller save*) und kopiert sie danach zurück.
2. Gerufene Prozedur rettet nach dem Unterprogramm-Aufruf diese Registerinhalte und kopiert sie vor dem Rücksprung zurück (*callee save*).

Sichern von Registern (2)

caller save

```
caller: Retten der Register  auf
        den stack.
        jal ...
callee:
        Retten von $31.
        Befehle f. Rumpf.
        Rückschreiben von $31.
        jr $31.
caller:
        Rückschreiben der
        Register.
```

callee save

```
caller: jal ...
callee:
    Retten der Register  auf
    den stack.
    Retten von $31.
    Befehle f. Rumpf.
    Rückschreiben von $31.
    Rückschreiben der
    Register.
    jr $31.
caller:
```

2.2.1.7 Prozeduren mit Parametern

```
int stund2sec(int stund)
{return stund*60*60}
stund2sec(5);
```

Wo findet stund2sec den Eingabeparameter?

Konflikt:

- Parameter möglichst in Registern übergeben (schnell)
- Man muss eine beliebige Anzahl von Parametern erlauben.

MIPS-Konvention:
Die ersten 4 Parameter werden in Registern $4, $5, $6, $7 übergeben, alle weiteren im *stack*.

Benutzung der Register

Register	Verwendung
$0	0
$1	Assembler
$2,$3	Funktionsergebnis
$4-$7	Parameter
$8-$15	Hilfsvariable, nicht gesichert
$16-$23	Hilfsvariable, gesichert
$24-$25	Hilfsvariable, nicht gesichert
$26-$27	Für Betriebssystem reserviert
$28	Zeiger auf globalen Bereich
$29	*Stack pointer*
$30	Zeiger auf Datenbereich
$31	Rückkehradresse

Call by value vs. *call by reference*

An Prozeduren Parameterwert oder dessen Adresse übergeben?

Parameterwert selbst (***call by value***):

Gerufener Prozedur ist nur der Wert bekannt.

Die Speicherstelle, an der er gespeichert ist, kann nicht verändert werden.

Adresse des Parameters (***call by reference***):
Erlaubt Ausgabeparameter, Änderung der Werte im Speicher durch die gerufene Prozedur möglich.
Bei großen Strukturen (*arrays*) effizient.

C: - Bei skalaren Datentypen (*int*, Zeigern, usw.): wählbar, typischerweise *call by value*,
- bei komplexen Datentypen (*arrays*): *call by reference*.

Prozedur swap ,
Prinzip der Übersetzung in Assembler

```
swap (int v[], int k)
{int temp;
    temp=v[k];
    v[k]=v[k+1];
    v[k+1]=temp;
}
```



Schritte der Übersetzung in Assembler:

1. Zuordnung von Speicherzellen zu Variablen
2. Übersetzung des Prozedurrumpfes in Assembler
3. Sichern und Rückschreiben der Register

Zuordnung von Registern

Variable	Speicherzelle	Kommentar
Adresse von v	$4	Aufgrund der MIPS-Konvention für den 1. Parameter
k	$5	Aufgrund der MIPS-Konvention für den 2. Parameter
temp	$15	(irgendein freies Register)
Interne Hilfsvariable	$16	(irgendein freies Register)

Realisierung des Rumpfes

```
temp=v[k];
v[k]=v[k+1];
v[k+1]=temp;
```

$4=
Adresse (v)




```
li   $2, 4          #
mul  $2,$2,$5       #Reg[2]:=4*k
add  $2,$4,$2       #Adresse von v[k]
lw   $15, 0($2)     #lade v[k]
lw   $16, 4($2)     #lade v[k+1]
sw   $16, 0($2)     #v[k]:=v[k+1]
sw   $15, 4($2)     #v[k+1]:=temp
```

Sichern der Register

Prolog: Sichern der in swap überschriebenen Register:

- Prolog ⬅
- Prozedurrumpf
- Epilog

```
addi $29,$29,-12
sw $2, 0($29)
sw $15, 4($29)
sw $16, 8($29)
```



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuerst** mittels **`addi`** den Platz auf dem *Stack* schaffen!

Sichern der Register

Prolog: Sichern der in swap überschriebenen Register:

- Prolog ⬅
- Prozedurrumpf
- Epilog

```
addi $29,$29,-12
sw $2, 0($29)
sw $15, 4($29)
sw $16, 8($29)
```



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuerst** mittels **`addi`** den Platz auf dem *Stack* schaffen!

Sichern der Register

Prolog: Sichern der in swap überschriebenen Register:

- Prolog ⬅
- Prozedurrumpf
- Epilog

```
addi $29,$29,-12
sw $2, 0($29)
sw $15, 4($29)
sw $16, 8($29)
```



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuerst** mittels **`addi`** den Platz auf dem *Stack* schaffen!

Sichern der Register

Prolog: Sichern der in swap überschriebenen Register:

- Prolog ⬅
- Prozedurrumpf
- Epilog

```
addi $29,$29,-12
sw $2, 0($29)
sw $15, 4($29)
sw $16, 8($29)
```



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuerst** mittels **`addi`** den Platz auf dem *Stack* schaffen!

Sichern der Register

Prolog
Prozedurrumpf 
Epilog

Rückschreiben der Register

Epilog: Rückschreiben der in swap überschriebenen Register:

- Prolog
- Prozedurrumpf
- Epilog ⬅

```
lw $2, 0($29)
lw $15, 4($29)
lw $16, 8($29)
addi $29,$29,12
```



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuletzt** mittels **addi** den Platz auf dem *Stack* freigeben!

Rückschreiben der Register

Epilog: Rückschreiben der in swap überschriebenen Register:

- Prolog
- Prozedurrumpf
- Epilog



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuletzt** mittels **addi** den Platz auf dem *Stack* freigeben!

Rückschreiben der Register

Epilog: Rückschreiben der in swap überschriebenen Register:

- Prolog
- Prozedurrumpf
- Epilog ⬅

```
lw $2, 0($29)
lw $15, 4($29)
lw $16, 8($29)
addi $29,$29,12
```



Wegen der Verwendung des *Stacks* auch für *Interrupts* (siehe 2.2.2.3) immer **zuletzt** mittels **addi** den Platz auf dem *Stack* freigeben!

Übersetzung von *bubble sort*

```
int v[10000]
sort(int v[], int n)
{int i,j;
  for (i=0; i<n; i=i+1) {
   for (j=i-1; j>=0 && v[j]>v[j+1]; j=j-1){
     swap(v,j);
   }
  }
}
```

Schritte der Übersetzung in Assembler:

1. Zuordnung von Speicherzellen zu Variablen
2. Übersetzung des Prozedurrumpfes in Assembler
3. Sichern und Rückschreiben der Registerinhalte

Zuordnung von Registern

Adresse von v	$4	Lt. Konvention
n	$5	Lt. Konvention
j	$17	gesichertes Register
Kopie von $4	$18	Sichern der Parameter zur Vorbereitung des Aufrufs von swap
i	$19	gesichertes Register
Kopie von $5	$20	wie $4
Hilfsvariable	$15,$16,$24,$25	irgendwelche freien Register
Hilfsvariable	$8	nicht gesichert

Übersetzung des Rumpfes

```
int v[10000]
sort(int v[], int n)
{int i,j;
 for(i=0;i<n;i=i+1){
  for(j=i-1;j>=0 &&
   v[j]>v[j+1];j=j-1)
   {swap(v,j);}}}
```

Adr. von v	$4
n	$5
j	$17
Kopie von $4	$18
i	$19
Kopie von $5	$20
Hilfsvariable	$15,$16,$24,$25
Hilfsvariable, nicht gesichert	$8

```
       add $18,$4,$0    #
       add $20,$5,$0    #
       li $19,0         # i:=0
for1:  bge $19,$20,ex1  # i>=n?
       addi $17,$19,-1  # j:=i-1
for2:  slti $8,$17,0    # j<0?
       bne $8,$0,ex2    #
       li $8,4          # $8:=4
       mul $15,$17,$8   # j*4
       add $16,$18,$15  # Adr(V[j]
       lw $24,0($16)    # v[j]
       lw $25, 4($16)   # v[j+1]
       ble $24,$25,ex2  # v[j]<=
       add $4,$18,$0    # 1.Param.
       add $5,$17,$0    # 2.Param.
       jal swap
       addi $17,$17,-1  # j:=j-1
       j for2           # for-ende
ex2:   addi $19,$19,1   # i:=i+1
       j for1           # for-ende
ex1 :  # ende
```

Sichern der Registerinhalte

Prolog: Sichern der über-schriebenen Register:

Prolog 

Prozedurrumpf

Epilog

Adr. von v	$4
n	$5
j	$17
Kopie von $4	$18
i	$19
Kopie von $5	$20
Hilfsvariable	$15,$16,$24,$25
Hilfsvariable, nicht gesichert	$8

```
addi $29, $29, -36
sw $15, 0($29)
sw $16, 4($29)
sw $17, 8($29)
sw $18, 12($29)
sw $19, 16($29)
sw $20, 20($29)
sw $24, 24($29)
sw $25, 28($29)
sw $31, 32($29)
```

Rückschreiben der Registerinhalte

Epilog: Rückschreiben der Registerinhalte:

Prolog

Prozedurrumpf

Epilog ⬅

Adr. von v	$4
n	$5
j	$17
Kopie von $4	$18
i	$19
Kopie von $5	$20
Hilfsvariable	$15,$16,$24,$25
Hilfsvariable, nicht gesichert	$8

```
lw   $15, 0($29)
lw   $16, 4($29)
lw   $17, 8($29)
lw   $18, 12($29)
lw   $19, 16($29)
lw   $20, 20($29)
lw   $24, 24($29)
lw   $25, 28($29)
lw   $31, 32($29)
addi $29, $29, 36
jr   $31
```

Zusammenfassung

- Nicht-verschachtelte Prozeduren
- jr- und jal-Befehle
- Das Stapel- (*stack*) Prinzip
- Verschachtelte Prozeduren
- Konventionen zur Registernutzung
- *Call-by-value*, *call-by-reference*
- Sichern und Rückschreiben von Registerinhalten
- *Callee-save* und *caller-save*
- Beispiele